

# SC\_Speclone

**SC\_Speclone** is a bonus utility program supplied free with **SC\_Assembler**. It allows a wide range of Spectrum 48K programs to run on Sam, even business and utility programs which use a printer.

Plus D disc snapshot files can be converted for use with Sam, and Spectrum 48K tape programs can be loaded and the whole Spectrum memory saved to disc to be re-loaded for future use. Only snapshots can be used from Plus D discs, not program files. So if you have a disc based Spectrum 48K program which you wish to use with Sam, and have no tape version, make a 48K snapshot using the Plus D. 128K programs and snapshots cannot be used with **SC\_Speclone**.

Appendix 2 contains technical information about the working of this utility.

On loading, the program goes to the main menu, illustrated below.

```
SC_SPECLONE @ STEVE'S SOFTWARE

1.RETURN TO SPECTRUM
2.RAND USR @ SPECTRUM
3.LOAD PLUS D SNAPSHOT FILE
4.CONVERT PLUS D SNAP FILES
5.SAVE SPECTRUM MEMORY
6.LOAD SPECTRUM MEMORY
```

## Option 1

### Return to Spectrum

Selecting this option will switch to Spectrum mode. If there is a Spectrum file in memory, this will be preserved.

### In Spectrum mode

All keys are scanned, but the following Sam keys have special uses:

**TAB** Graphics mode - equivalent CS/9

**CNTRL E**-mode - equivalent CS/SS

**ESC BREAK** - equivalent CS/SPACE

## **FUNCTION KEYS 0-9 Keypad returning the digits 0-9**

INV hash mark #

### **Printing from Spectrum mode**

LPRINT sends ASCII characters to the printer, with tokens unexpanded. For example, LPRINT "Testing" will work. LLIST also sends only ASCII characters and unexpanded tokens, so it is not possible to LLIST a Spectrum program, because the keyword codes will not be converted to spell out the keywords.

### **Linefeeds**

The program is set up to send a linefeed after each carriage return. If your printer already sends a linefeed automatically, the result will be permanent double line spacing. To correct this, the Spectrum ROM file on the **SC\_Speclone** disc must be modified.

Reset Sam, after saving the Spectrum memory if necessary. Put the **SC\_Speclone** disc in drive 1, and enter the following lines as direct commands

```
CLEAR 32767
LOAD "rom" CODE
POKE 80324,195
SAVE OVER "rom" CODE 65536,16384
```

### **Printing to #3**

Most Spectrum utilities do not use LPRINT, but send the bytes to be printed to be output to #3 via a printer driver routine. **SC\_Speclone** has a suitable printer driver at Spectrum addresses 14793-14826 (34 bytes) (Sam addresses 80329-80362). This routine should be copied to the address at which the Spectrum program's own printer driver routine resides, overwriting the program's routine.

### **Spectrum addresses from Sam BASIC**

From Sam BASIC, all Spectrum addresses reside 65536 bytes higher than their normal Spectrum address. (See memory map in Appendix 2).

### **Switching to Sam from Spectrum mode**

#### **NEW**

The Spectrum keyword NEW returns to Sam. If Spectrum address 23296 holds 0, the return will be to the main menu. If 23296 holds any other value, the return will be to Sam BASIC line 1000.

### **NMI button.**

If the Spectrum program does not permit you to exit to BASIC, the NMI

button can be used to return to Sam BASIC, but Sam's NMI button has a fault which needs a hardware modification. It will exit to Sam BASIC, but you should not rely upon being able to return to the Spectrum program at the point from which you left it.

### **To NEW the Spectrum memory.**

The keyword NEW cannot be used in the usual way, because it is used to switch to SAM mode.

### **PRINT USR 14888**

Mimics NEW. BASIC programs are cleared from memory, but code stored above a CLEAR address is preserved.

### **PRINT USR 0**

Resets Spectrum, clearing all programs and code from memory.

### **Saving and loading from Spectrum mode**

In Spectrum mode, SAVE and LOAD will normally be to tape, and the program compensates automatically for tape loading speed. Disc and microdrive syntax are not accepted. Disc saving and loading of code blocks must be done by storing variables such as file start and length in Spectrum memory, switching to Sam, retrieving the variables and saving the code block from Sam BASIC.

### **Example of conversion of Spectrum program for file saving to disc.**

The Spectrum variable 23296, which controls the method of returning to Sam BASIC, will hold 0 if the return is to the main menu. If it holds any other number, the return will be to Sam line 1000. In Sam BASIC, Spectrum addresses lie 65536 above their working Spectrum address, and so the contents of this variable can be retrieved from Sam BASIC by PEEK(65536+23296).

A typical Spectrum microdrive SAVE routine is

```
5000 GOSUB 6000:SAVE * "M";D;F$ CODE start,length:RETURN
6000 LET D=PEEK 32768:LET start=PEEK 32769+256*PEEK 32770:LET
length=PEEK 32771+256*PEEK 32772:LET F$="":FOR A=0 TO 9:LET
F$=F$+CHR$ PEEK(32773+A):NEXT A:RETURN
```

To convert this to saving from Sam, the Spectrum BASIC must be

```
5000 POKE 23296,1:NEW:RETURN
delete line 6000
```

23296 holding 1 will force the jump to Sam line 1000 when NEW switches to Sam. The Sam command GO TO 1 will return to Spectrum mode to execute the instruction immediately following NEW. 23296 could be made to hold 1 for SAVE, 2 for LOAD, 3 for DIR, 4 for ERASE and so on. The necessary Sam BASIC will be

```
1000 LET N=PEEK(65536+23296)
1010 IF N=1 THEN SAVER:ELSE IF N=2 THEN LOADER:ELSE IF N=3 THEN
DIRE:ELSE IF N=4 THEN ERASER:END IF
5000 LABEL SAVER:GOSUB 6000:SAVE "D"+CHR$(D+48)+":"+F$ CODE
start+65536,length:GO TO 1
6000 LET D=PEEK(32768+65536):LET start=DPEEK(32769+65536):LET
length=DPEEK(32771+65536):LET F$=MEM$(32773+65536 TO
32773+9+65536):RETURN
```

Line 1000 will fetch the contents of the Spectrum variable 23296, and line 1010 will call the appropriate Sam subroutine to perform the correct disc operation. 23296 holding 1 would call the save routine at line 5000. Line 6000 mimics the Spectrum line 6000, but peeks addresses 65536 above the Spectrum ones and uses Sam's more economical syntax. After calling line 6000, line 5000 saves the required code block to disc and GO TO 1 returns to Spectrum mode to execute the RETURN which follows NEW in Spectrum line 5000.

Similar subroutines could be written to perform the other disc operations. Note that Sam labels cannot be keywords such as SAVE or DIR - another letter must be added. See also Appendix 3.

### **SC\_Speclone menu option 2 RANDOMIZE USR 0 SPECTRUM**

This option returns to Spectrum mode and clears the Spectrum memory. It should be used the first time a jump is made to Spectrum mode, unless a Spectrum program has already been loaded.

### **Menu option 3 LOAD PLUS D SNAPSHOT FILE**

This option should only be used to load Plus D snapshots which have previously been converted using option 4.

If, when a converted snapshot is loaded, there is no response to the keyboard on returning to Spectrum mode, use the NMI button to return to

the Sam mode menu and follow this procedure.

Press ESC - goes to Sam BASIC.

Enter POKE 80290,195

Enter GO TO 10 - returns to the SC\_Specclone menu

Use option 3 and re-load the snapshot.

Most snapshots are compatible with the program, and this POKE will enable the keyboard response for the majority. If you have a snapshot which requires this POKE, in future exit to Sam BASIC and do the POKE before loading the snapshot.

#### **Menu option 4**

#### **CONVERT PLUS D SNAP FILES**

This option must be used before a snapshot can be run under SC\_Specclone. If the POKE described above has been used to enable the keyscan of another snapshot, it must be restored before using this option.

Press ESC to go to BASIC.

Enter POKE 80290,226

Enter GO TO 10 to return to menu.

Put the disc containing the Plus D snapshot into drive 1 and select option 4. The catalogue will be displayed, and you will be prompted for the file number of the program to be converted. After a brief pause while the conversion is made, you will be prompted for the filename under which it is to be saved. Put the disc on which you wish to save it in drive 1 and give the filename. After saving, the program returns to the main menu. Option 3 may be used to load the converted snapshot.

#### **Menu option 5**

#### **SAVE SPECTRUM MEMORY**

The complete Spectrum memory, from Spectrum addresses 0-65535 is saved to disc.

#### **Menu option 6**

#### **LOAD SPECTRUM MEMORY**

Loads files saved under option 5. After loading, select menu option 1 to return to Spectrum BASIC with the memory preserved.

## SC\_Speclone Memory map

Page	Sam addresses	Used for
0	16384-32767	Sam memory
1	32768-49151	Sam memory
2	49152-65535	Sam memory
3	65536-81919	Spectrum 48K ROM 0-16384
4	81920-98303	Spectrum memory 16384-32767 includes Spectrum screen
5	98304-114687	Spectrum memory 32768-49151
6	114688-131071	Spectrum memory 49152-65535

### How SC\_Speclone works

The Spectrum 48K ROM is modified to scan for the extra Sam keys such as DELETE and the function keys. This code is placed in a free area of Spectrum memory between 14446 and 15615. Bytes 11446-14893 are used for the keyscan and other essential code. All Sam keys are scanned. See pages 25-26 for the special uses assigned to some of the keys.

A printer driver routine is proved at Spectrum addresses 14793 to 14826. (34 bytes). This enables the LPRINT command. LLIST mimics LPRINT because tokens are not expanded by this routine. The printer driver sends a linefeed after every carriage return. Instructions for disabling the linefeed are on p.26.

An OUT instruction is used to page the Spectrum ROM to Sam address 0 and to use screen MODE 1, the Spectrum compatible mode, at Spectrum address 16384 when switching to Spectrum mode.

The Spectrum NEW command is used as a switch to return to Sam mode, paging out the Spectrum ROM and paging in the Sam ROM and setting screen 1, MODE 4, at the normal Sam screen pages, the last two pages in memory. RANDOMIZE USR 14888 mimics the normal Spectrum NEW while in Spectrum mode.

In Sam mode all Spectrum addresses reside 65536 above their normal Spectrum address, and so a Spectrum address  $n$  may be POKEd from Sam BASIC at address  $n+65536$ .

In Sam BASIC GO TO 10 returns to the main menu.

GO TO 1 returns to Spectrum mode without resetting Spectrum memory.

**SC\_Speclone example**  
**PCG's DTP PACK conversion**

If you have a disc-based or microdrive version of Spectrum DTP PACK, you must first make a tape copy of the "WM" code block. Reset the Spectrum. Enter CLEAR 24733; LOAD \*m";;"WM" CODE 54174. When the code block has loaded, enter SAVE "WM" CODE 54174,11362 and save the code block to tape. Prepare a newly formatted disc with only the SAMDOS file on it.

Now, using Sam, load the **SC\_Speclone** utility and select menu option 2 - RANDOMIZE USR 0 SPECTRUM. In Spectrum mode, enter CLEAR 24733;LOAD "WM" CODE and play the tape to load the code block.

Type in the following lines of Spectrum BASIC.

```
10 LET D=NOT PI:LET S=D:LET L=D:LET X=D:LET A$="          ":RANDOMIZE
USR 63315
20 POKE 23296,1:GOSUB 60:NEW:RANDOMIZE USR X
30 POKE 23296,2:GOSUB 60:NEW:RANDOMIZE USR X
40 POKE 23296,3:GOSUB 60:NEW:RANDOMIZE USR X
50 POKE 23296,4:NEW:RANDOMIZE USR X
60 LET V=INT (S/256):POKE 23297,S-(256*V):POKE 23298,V
70 LET V=INT (L/256):POKE 23299,L-(256*V):POKE 23300,V
80 FOR A=1 TO 10:POKE 23300+A,CODE A$(A):NEXT A
90 RETURN
100 POKE 23296,5:NEW:RUN
200 POKE 23296,0:NEW:RUN
300 POKE 65532,158:POKE 65533,96:RUN
```

In line 10, there are 10 spaces in A\$. Lines 20 to 50 POKE a code into 23296, to tell Sam BASIC which DOS operation to perform, and they replace the LOAD, SAVE, ERASE and CAT lines of the original Wordmaster BASIC. The subroutine at 60 pokes the file start and length and filename into variables from which Sam BASIC can retrieve them. Line 100 is a line which will return to Sam BASIC, and line 200 to the **SC\_Speclone** menu. Line 300 will clear all files from Wordmaster's memory, and return you to the program, providing a quick way of deleting multiple files. The DOS commands are called normally, from the program's options. To use lines 100-300 you must exit from the program to Spectrum BASIC.

Now enter GO TO 200, put the prepared disc in drive 1, and select option 5 - SAVE SPECTRUM MEMORY. Give the file name "WM" when prompted. When the Spectrum memory has been saved, press ESC to return you to Sam BASIC.

Type in the following lines of Sam BASIC.

```
1000 LET A=PEEK (23296+65536):ON A:GO TO LOADER:GO TO SAVER:GO TO
ERASER:GO TO CATTER:STOP
1200 DEF PROC GETVARS
1210 LET S=256*PEEK (65536+23298)+PEEK (65536+23297)
1220 LET L=256*PEEK (65536+23300)+PEEK (65536+23299)
1230 DIM A$(10)
1240 FOR A=1 TO 10:LET A$(A)=CHR$ PEEK (65536+23300+A):NEXT A
1250 END PROC
1500 LABEL LOADER
1510 GETVARS:LOAD A$ CODE S+65536,L:GO TO 1
1600 LABEL SAVER
1610 GETVARS:SAVE A$ CODE S+65536,L:GO TO 1
1700 LABEL ERASER
1710 GETVARS:ERASE A$:GO TO 1
1800 LABEL CATTER
1810 CLS:DIR:PAUSE 0:GO TO 1
8000 OPEN##5,"b":PRINT ##5,CHR$ 27;"C";CHR$ 70;:CLOSE ##5:RETURN
```

Lines 1000 to 1810 perform the DOS operations. Line 1000 PEEKs the variable to discover which operation is required and directs the program to the correct subroutine. Each operation ends in GO TO 1, which returns to the Spectrum program at the command after NEW, which switched to Sam mode. The procedure at 1200 sets up the variables. Line 8000 contains any codes you may wish to send to the printer. The ones given set up A4 paper length, but they may be changed to any you wish to use. If you use the line to send printer codes, the printer must be on line when you load the program.

Now alter line 9000 to read

```
9000 CLS ##:PALETTE##:CLS ##:CLEAR 29999:LOAD "rom" CODE:LOAD "high"
CODE:LOAD "low" CODE:LOAD "WM" CODE:DPOKE (23730+65536),24733:GO
SUB 8000:POKE 88832,0:RUN 1.
```

You have added commands to load the Spectrum memory, POKE the Spectrum RAMTOP and call the printer codes subroutine, and changed the RUN address to 1. You can now save the Sam BASIC to your prepared disc.

SAVE "AUTOWORD" LINE 9000.

Now enter POKE 80324,195 and SAVE "rom" CODE 65536,16384. Finally, you must copy the "high", and "low" code blocks from your SC\_Speclone disc to your new disc. The program will autoloading, and will be exactly like the Spectrum version except that the tape load/save operations are unuseable. PlusD disc files may be used, but if you have the tape version of DTP PACK, you must copy all the extension programs, fonts, etc. to Sam discs.